

UNITED STATES DEPARTMENT OF THE INTERIOR
GEOLOGICAL SURVEY

HP-85 Computer Programs for Reading Sprengnether DR-100
Seismograph Cassettes with DP-100 Playback System

John R. Evans¹

Reese Cutler²

Open File Report 83-521

This report is preliminary and has not been reviewed for
conformity with U. S. Geological Survey editorial standards.
Any use of trade names is for descriptive purposes only and
does not imply endorsement by the U. S. Geological Survey.

¹Menlo Park, California

²Cutler Digital Design
Mountain View, CA

June 1983

Introduction

This Open-File report describes BASIC language and HP-85 Assembly Language computer programs which allow a Hewlett-Packard HP-85 microcomputer to read seismic data from a Sprengnether DP-100 playback unit. The DP-100 plays back digital cassettes recorded on Sprengnether DR-100 digital seismographs and can operate at 2, 5, or 10 inches per second (ips). The BASIC language program "DPLLOT" uses the assembly language routine "DPB" to transfer data from the DP-100 to the HP-85 at up to 10 ips. A maximum of 9000 samples (including an 8-word header) can be transferred to the HP-85 (30 seconds of seismogram for 3 channels, 100 samples per second each). No provision is made for transferring larger files beyond the first 9000 samples.

Tranferred seismograms can be stored on the HP-85 tape cartridge, processed for phase times and amplitudes, plotted on the computers' printer, or any combination of the above. Seismograms stored on the HP-85 cartridge can be processed or plotted later as well.

The remainder of this report is comprised of the programmer's operation manual (somewhat edited) and program listings. I have added notes (in double parentheses) where appropriate. In particular note that a constant number "11" must be changed to a constant number "9" on lines 2310 and 4020 in the listing of program DPLLOT when used with newer models of the DP-100 (that is, models with correct duration of the word identification bits--bits 12 and 13--at the DP-100 output connector).

The major limitation of this system is the speed of the plotting software--plotting a 9000-sample record takes over 10 minutes. Picking takes longer because it begins with a plot of the whole event, followed by plots of one or two smaller sections of seismogram in which the picks are actually made.

Because the plotting is slow and because the HP-85 has a slow microprocessor this system is recommended for field use where other computers are not available. If high-speed (9600 baud) communication software were available, the HP-85, acting as a computer terminal, could serve as a reasonably-efficient DP-100-to-main-computer interface since plotting is not needed for transferring records. At present, only 300-baud communication software is known to exist.

Extensions to this software package are available to the public* for automatic record transfer and automatic plotting of entire DR-100 tapes. With these extensions the operator need be present only periodically and can do other work in the meantime.

*Write to the first author at 345 Middlefield Road, MS-77, Menlo Park, CA 94301.

DP-100 SERVICE PROGRAM
ON HP-85

Cutler Digital Design
2215 Sun Mor Ave.
Mountain View, California 94040
(415)964-1481

This document describes the HP-85 BASIC program, DPLOT, that accesses, stores and plots time sequence data from the Sprengnether DP-100 play back unit.

DPLOT

The HP-85 computer program DPLOT is used to control and read the Sprengnether DP-100 play back unit. The program can store up to 9,000 data points, which is equivalent to data sampled from three channels for thirty seconds at 100 samples per second. Data can be stored on an HP-85 tape file, and then will be available for plotting, or for identification of specific points according to time.

To be able to run DP-100 the user needs an HP-85 with a 16K memory module, the ROM drawer with the I/O ROM in it, and the GPIO interface module with proper switch settings and cable preparation.

OLD DP-100 VERSUS NEW DP-100

The program, DPLOT has been developed using an early version of the DP-100. Because of the hardware problems of this particular generation of DP-100 units, the software has special patches which insure that data are related to the appropriate channels. If a newer unit is being used (e.g. if the user's unit allows time marks to be removed) the user will have to alter the DPLOT program.

HARDWARE SETUP

Before using DP-100, the user must insure that the HP-85 is properly configured with an I/O ROM (HP part number 00085-15003), a ROM drawer (HP part number 82936A), a 16k memory module (HP part number 82903A), and GPIO interface (HP part number 82940A) with proper cables. Also, the DP-100 playback unit should have its switches set correctly.

The I/O ROM should be plugged into one of the six sockets of the ROM drawer. Then the ROM drawer should be plugged into one of the I/O slots in the back of the HP-85.

The memory module should be plugged into another I/O slot in the back of the HP-85.

The GPIO interface module must have switches properly set inside, and a connector attached to the two cables that come out of it. It is recommended that the two be cut to equal length. Use the pin to color-coded wire assignments in the appendix to see how to connect the cables to the standard male DB25P plug that mates with the DP-100 socket. The manual that comes with the GPIO interface has a good discussion of cable preparation.

The switches inside the GPIO interface module should be set as follows:

EIGHT SWITCH PACKAGE

POSITION 1 - OFF

POSITION 2 - OFF

POSITION 3 - ON

POSITION 4 - OFF

POSITION 5 - ON

POSITION 6 - OFF

POSITION 7 - ON

POSITION 8 - ON

FOUR SWITCH PACKAGE

POSITION 1 - ON

POSITION 2 - ON

POSITION 3 - ON

POSITION 4 - OFF

The DP-100 should be set to a playback speed of 10 inches per second and it should be placed in single event mode.

PROGRAM STARTUP

To begin to use DPLOT, the user must configure the hardware as described above. The user then turns on the HP-85 and waits for the cursor (the line segment in the upper left hand corner) to appear.

Next, the user loads the program by putting a program tape in the HP-85 tape drive and typing:

LOAD "DPLOT"

followed by striking the "endline" key. It will take approximately one minute to load. When it is ready, the tape drive light will go out, and the display, which goes blank during tape operations, will return.

The user starts the program, DPLOT, by striking the "run" key ((program tape must still be in drive at this point)). It will take anywhere from a couple of seconds to a minute for the program to return the following display*:

*Throughout this document the displayed characters and graphics are referred to as a "display", "screen", "graph" or "plot".

COMMAND SUMMARY:

```

STOP
REWIND
FASTFORWARD
TRANSFER
DIRECTPLOT
RECORDPLOT
PICKDUMP
AUTO
?
```

((change to data tape now if you wish))

This display is called the master menu. It indicates what types of operations can be specified, and what must be typed to invoke them (make them happen). There are two groups of operations: tape movement and data handling.

TAPE MOVEMENT COMMANDS

It is easy to understand the tape movement instructions. With the DPLLOT program loaded and running, the user turns on the DP-100, and inserts a DR-100 data tape in the DP-100 tape drive. The user may position the tape using the DP-100 controls or the following tape movement commands.

To rewind the tape, the user must type "REWIND" followed by striking the "endline" key.

To make the tape advance in the fast forward mode, the user must type "FASTFORWARD" followed by striking the "endline" key.

To halt a fast forward or a rewind operation before the end of the tape, the user must type "STOP" followed by striking the "endline" key.

DATA HANDLING COMMANDS

The six general operations in the data handling commands are: 1) getting data from the DP-100, 2) storing data on the HP-85 data tape, 3) getting data from the HP-85 data tape, 4) plotting data on the display and printer with a printed header, 5) allowing the user to interactively pick certain points in the data record and have them stored in a separate data file (a pick file), and 6) displaying the contents of a pick file.

Here is a brief description of the four data handling commands - notice that the first three all allow plotting or point picking after data is acquired:

TRANSFER reads the next data record from the DP-100 (wherever the tape is positioned on the DP-100). The data is stored on a HP-85 data tape and then the user is allowed to proceed to plotting and picking if desired.

DIRECTPLOT reads the next data record from the DP-100 (wherever the tape is positioned on the DP-100). Without storing the data, the user proceeds to picking and plotting.

RECORDPLOT reads the data record from an HP-85 data file and proceeds to picking and plotting.

PICKDUMP provides a listing of the data points saved in a pick file.

TRANSFER

To invoke the transfer command, the user simply types "TRANSFER" followed by an endline. The HP-85 will start the DP-100 and will collect a record of data (up to 9000 data points). When a complete record is received by the HP-85 it will prompt with:

```
FILENAME
?
```

The user responds by typing the HP-85 file name that is to be used for this data. If the file name is acceptable (i.e. it has not already been used for a file on this particular tape) and if there is enough room on the tape, the data will be recorded on HP-85 tape. When the data transfer has been completed the HP-85 will display the following prompt on its screen:

```
PLOT(Y/N)?
```

To try plotting or point picking, the user must type "Y" and an endline. Typing anything else (and an endline) will terminate the operation and return the user to the master menu.

If there was an error in file specification, the computer will list the catalog of files in the current data tape and allow the user to re-enter the file name (or change data tapes).

For instructions on plotting and picking see the section below.

DIRECTPLOT

To invoke the direct plot command, the user simply types "DIRECTPLOT" followed by an endline. The HP-85 will start the DP-100 and will collect a record of data (up to 9000 data points). When a complete record is received by the HP-85, the program will proceed to the plotting and picking section which is described below.

RECORDPLOT

To invoke the record plot command, the user simply types "RECORDPLOT" followed by an endline. The HP-85 will prompt:

FILENAME
?

The user responds by typing the name of the data file followed by an endline. The HP-85 will retrieve the data record from the data file (this could take some time) and then proceed to the plotting and picking portion of the program.

If the user gives the name of a file that does not exist, the HP-85 will respond with a listing of the catalog of the current data tape and will allow the user to enter the file name again.

PICKDUMP

To invoke the pick dump command, the user simply types "PICKDUMP" followed by an endline. The HP-85 will respond with the following prompt:

SPECIFY FILE WITH STORED POINTS
FILE?

The user types in the name of the picked point file followed by an endline. The program will access the file, display how many point sets exist in the pick file, and how many are possible. Finally the program will print the following pick file information for each pick set:

- 1) Descriptor string
- 2) Point A information
- 3) Point B information
- 4) Point C information
- 5) Point D information

6) Which DR-100 the data was recorded on

7) Which channel the data was picked from

When the printing is done, the program returns to the master menu. If there was a problem with the specified file name, the program displays a catalog of the files on the current HP-85 data tape and allows the user to re-specify the file name.

EXAMPLE OF PICKDUMP OUTPUT

```
2 SETS OF PICKED PTS
10 POSSIBLE
ON FILE - POINTS
```

TEST RECORD

```
A 13479706.9433 .903 PTA
B 13479706.9133 .820 PTB
C 13479706.9932 .879 PTC
D 13479707.0132 .869 PTD
DR100 # 202
CHANNEL 1
```

CHANNEL ONE EXAMPLE PICK

```
A 13479708.1246 .034 POINT 1
B 13479708.5040 -.356 POINT 2
C 13479708.8634 .903 POINT 3
D 13479709.3127 -.542 POINT 4
DR100 # 202
CHANNEL 1
```

((The line before the picks is a general comment about the set of picks. The long number in field two is seconds after the beginning of the year. The smaller number in field three is the amplitude of the trace at the pick time (volts). The characters in field four are arbitrary pick labels up to 10 characters long.))

PLOTTING AND PICKING

When the user comes to the plotting and picking portion of the program the HP-85 will display the following prompt:

PLOT,PICK, OR EXIT?

If the user wishes to stop plotting or picking at this point, s/he simply types "STOP" and an endline and the program will return to the master menu.

If the user wishes to plot or pick, simply type "PLOT" or "PICK" followed by an endline. When the user is finished with the

plotting or picking, the program will return to this section and choose again. See below for the specifics of plotting or picking.

PLOT

When the user responds with "PLOT" to the "PLOT, PICK OR EXIT" query, the HP-85 will ask for the magnification:


MAGNIFICATION?

DR-100/DP-100 data has a possible data range of plus or minus ten volts for each data sample. When the actual data value movement is very small, a plot of the full allowable range might not show very much. The user may, therefore, wish to magnify the data. This is done by typing a real number. If the number is greater than one, the waveform of the data will get larger.

After the user has specified the magnification, the HP-85 will ask for time scale information:

SECONDS PER CM?

A standard value is 1. The user must type the appropriate real number and endline. The plot of all data channels and a header will be produced at the HP-85 printer:



BACKGROUND FOR PICKING POINTS

Before proceeding with the steps involved in picking points a little discussion is necessary.

One peculiarity of DR-100 data is that the exact timing corresponds to timing marks that may appear at many different positions in the data set. While these marks are usually evident to the human eye, they are not as easy to spot for a computer program. Hence, for any DR-100 data set that is recorded on HP-85, the first step in picking points is to identify the timing marks. To increase the efficiency of the picking operation, the timing mark is selected in the following sequence:

- 1) The plot of the whole data record is made ((not just a first window as in "plot")).
- 2) Using cursor controls, the user specifies a window around the timing mark he is going to identify.
- 3) (Optional) The user can specify a window around a second area of interest in the data record (called the interest window).
- 4) The user identifies the timing mark in the timing mark window.
- 5) The user proceeds with picking points in the interest window.

Once a timing mark has been identified for a data set that is recorded on HP-85 data tape, it is remembered, and therefore it will not be necessary to go through the identification process again.

It is important at this time to talk about special keys on the HP-85 keyboard. Besides the normal typewriter keys, the HP-85 has many special keys. In this program, two sets of them are particularly useful: cursor movement keys and special function keys.

SPECIAL FUNCTION KEYS

The following plot appears at one point in the point picking sequence. It shows labels for the special function keys (described below).

YOU MUST SPECIFY A TIME MARK
 WINDOW
 YOU MAY SPECIFY AN INTEREST
 WINDOW

```

-----
CONT    GRAPH
[TIME   TIME]  [    ]

```

The special function keys are the five above the standard keyboard that are labelled: keylabel, k5/k1, k6/k2, k7/k3 and k8/k4. Pressing the "keylabel" key displays a short description of the current function of the other four keys. These functions change throughout the program. There are eight possible functions, for the four special function keys can be used in combination with the shift key. In the previous display there is a line of dashes and two lines of characters at the bottom. The six positions in those two lines of characters describe the six possible functions that the special keys have at this part of the program. (Two of the possible eight functions are not used now.) For example, at this juncture a plot has been drawn, but it is not currently displayed. Invoking the "GRAPH" function displays whatever plot is in the computer. From the key labels we see the word GRAPH in the second position on the top. This means that pressing k6/k2, while holding the shift key down, will invoke the "GRAPH" operation. This will produce the plot. Pressing the keylabel key causes the text to return. Before explaining the meaning of the other five functions we need to discuss the cursor control keys.

CURSOR CONTROL KEYS

In some plots of seismograms an arrow will appear (e.g. page 14). This arrow is called the graphics cursor. The point that it indicates is at the tip of the arrow. The user can move the graphics cursor around the screen by using the arrow buttons at the top right of the HP-85 keyboard. While pressing the buttons normally causes a small movement in the direction of the arrow of the particular button, the user can speed up the movement by holding the shift key down before striking the arrow button. Where the cursor crosses data points, the intersection of the cursor and the data point is not illuminated. This fact can be helpful in

proper cursor placement. Also, if the screen is not in "GRAPH" mode (i.e. there is just text on the screen), pressing a cursor movement arrow switches the screen to "GRAPH" mode.

PICK

When the user responds with "PICK" to the "PLOT, PICK OR EXIT" query, the HP-85 will prompt:

SPECIFY FILE FOR SAVING POINTS
FILE?

The user responds with the name of a pick file. If a new file is specified, the HP-85 will request file length information:

NEW FILE

NUMBER OF SETS OF PICKED POINTS
IN NEW FILE?

((There are 4 picks in each set and one set is usually sufficient. For example, P, S, and a pick for measuring signal duration will fit in one set with room to spare.)) Once the file is set up, the user will be asked which channel is to be the source of points:

CHANNEL (1,2,3,ALL)?

Finally, before picking begins, the user will be asked for a magnification factor:

MAGNIFICATION?

Now the program has all the information it needs to begin plotting data. It performs the plot. When the plot is completed, the following text display will appear on the HP-85 screen (if plotting from an HP-85 tape file skip the following section):


```

      YOU MUST SPECIFY A TIME MARK
      WINDOW
      YOU MAY SPECIFY AN INTEREST
      WINDOW
  
```

```

-----
CONT    GRAPH
[TIME  TIME]  [    ]
  
```

To display the plot on the screen, invoke the "GRAPH" function by pressing the k6/k2 key while holding the shift key down. This will produce:





Before picking points the user must identify a time mark. To do this the user must position the cursor to the right of the timing mark to be identified and invoke "TIME]". Then the user must position the cursor to the left of the timing mark to be identified and invoke "[TIME".

If desired, the user may use the "[" and "]" functions to specify an interest window to be returned to after the time mark is identified. If the user does not specify an interest window, the whole plot will be re-drawn after the time mark is identified. ((User should identify an interest window around the phase of interest to reduce plotting time and increase pick accuracy.))

When both time window boundaries are specified the user may invoke the "CONT" function to continue on with the program. The time mark window will then be plotted and the following message will appear:

POSITION CURSOR FOR TIME MARK
AND HIT K1 WHEN READY

To pick a time mark (or other point) the user must imagine a vertical line intersecting the plot at that point. The user must place the tip of the cursor arrow anywhere on that line. Finally, the user must press the k5/k1 key (without shifting). Below is an example of the plot before and after moving the cursor to the timing mark:




Once the user presses the k5/k1 key the following text appears:

TYPE THE INTEGER FOR THE TIME
MARK YOU HAVE SELECTED
1 - FIRST TIME MARK
2 - SECOND TIME MARK
X - XTH TIME MARK

The user must now tell the computer which time mark was indicated. ((From the event's header on the DR-100 cassette, the system knows the time of the first mark and that subsequent marks appear every other second thereafter. There is no provision for DR-100 clock corrections, but you can compensate when picking the time mark.))

When this has been done, the program will proceed to plot the interest window. Again it will ask the user to specify a channel and a magnification. Then it will return with a plot like the following:



Pressing the keylabel key gives the special functions that are allowed at this point of the program:

```
-----
STOP      ORGPLOT      MARK
GRAPH     REPLOT      COPY
```

STOP exits the "PICK" part of the program.

GRAPH puts the screen in graph mode.

ORGPLOT re-plots the data including the whole data record on the screen. Channel and magnification queries will follow.

REPLOT re-plots the data according to a new interest window boundary as set by the "[" and "]" functions under MARK. Channel and magnification queries will follow.

COPY produces a printed copy of the current plot with a text header.

MARK assumes that the cursor has been placed in a meaningful position, and that one of seven types of marks is to be noted, or that previously marked data is to be stored in the pick file.

When the user invokes MARK (presses the k8/k4 key with the shift key held down) the following display appears:

TYPE?

```

-----
[      SAVE      TIME      ]
PT A    PT B    PT C    PT D
-----

```

The types of marks are:

The "[" function sets a new left hand boundary for the interest window. This boundary is not used until a REPLOT function is invoked.

The "]" function sets a new right hand boundary for the interest window. This boundary is not used until a REPLOT function is invoked.


The "TIME" function resets the time mark according to the current cursor position. ((This function works (we think) but is archaic since the time mark is usually read as described in pages 13-16.))

The "SAVE" function stores all picked points on the tape file. This function must be invoked before leaving the "PICK" portion of the program in order to have a record of picked points.


The "PT A", "PT B", "PT C" and "PT D" functions mark points A, B, C, and D. It is this set of points that is saved during a SAVE operation. Invoking one of these functions causes the program to mark the point and ask the user for a short string descriptor (ten characters or less) of the point.

POINT PICKING - USER SEQUENCE

After setting the time mark and replotting around the interest window (which will happen when entering the PICK portion of the program), the plot will look something like this:



The user moves the cursor to the position of the first point of interest using the arrow keys:



The user now invokes the MARK function by striking the k8/k4 key while holding the shift key down. The display will show:

TYPE?

```

-----
[   SAVE   TIME   ]
PT A  PT B  PT C  PT D

```

The user now strikes one of the function keys to indicate the type of point picked (for example, k5/k1 to pick point A). The computer then asks for a short descriptive name of the point:

```

TYPE?
POINT NAME
?


```

```

-----
[   SAVE   TIME   ]
PT A  PT B  PT C  PT D

```

The user types a short label for the point (e.g. FB or EP-2), followed by striking the endline key. The plot will return with a notation of where the picked point is:




Notice that now, if the user presses the keylabel key there is also a notation for point A:

A 13479708.1246

STOP	ORGPLOT		
GRAPH	REPLOT	COPY	MARK

Marking points B, C and D similarly produces a plot:



Now, the user can save these points by invoking the MARK function and then invoking the SAVE function when the computer prompts for pick type. After you indicate SAVE the display will show:

```
TYPE?
RECORD DESCRIPTOR ?
```

```
-----
[      SAVE      TIME      ]
PT A    PT B    PT C    PT D
```

The user types in a description of the pick set followed by striking the endline key. The pick set has been saved, and the program is ready for another set of picks or another seismogram.

BASIC LISTING OF "DPLLOT"

((There are two places where a constant "11" must be changed to a constant "9" if this program is used with new-model DP-100's. See lines 2310 and 4020.))

```

10 RESET 4
20 CONTROL 4,4 ; 192+4,10
30 OUTPUT 4 USING "#,K" ; CHR$(
0)&CHR$(0)
40 A=0 @ ON ERROR GOTO 70
50 A=DPC(CHR$(15)&CHR$(255))
60 IF A=10 THEN 90
70 OFF ERROR
80 LOADBIN "DPB"
90 OFF ERROR
100 DIM A$(18016),C$(20),B5$(64)
,B9$(64)
110 D8=18016
120 DATA 1536
130 DATA 768
140 DATA 384
150 DATA 128
160 DATA 3072
170 DATA 1536
180 DATA 768
190 DATA 384
200 DATA 3072
210 DATA 1536
220 DATA 768
230 DATA 384
240 GOSUB 260
250 GOTO 530
260 ON KEY# 1,"GRAPH" GOSUB 560
270 ON KEY# 2,"RELOT" GOSUB 570
280 ON KEY# 3,"COPY" GOSUB 580
290 ON KEY# 4," " GOSUB 590
300 ON KEY# 5,"STOP" GOSUB 600
310 ON KEY# 6,"ORGPLOT" GOSUB 61
0
320 ON KEY# 7," " GOSUB 620
330 ON KEY# 8,"MARK" GOSUB 630
340 RETURN
350 ON KEY# 1,"[TIME" GOSUB 560
360 ON KEY# 2,"TIME]" GOSUB 570
370 ON KEY# 3," [" GOSUB 580
380 ON KEY# 4," ] " GOSUB 590
390 ON KEY# 5,"CONT" GOSUB 600
400 ON KEY# 6,"GRAPH" GOSUB 610
410 ON KEY# 7," " GOSUB 620
420 ON KEY# 8," " GOSUB 630
430 RETURN
440 ON KEY# 5," [" GOSUB 600
450 ON KEY# 6,"SAVE " GOSUB 610
460 ON KEY# 7,"TIME " GOSUB 62
0
470 ON KEY# 8," ]" GOSUB 630
480 ON KEY# 1,"PT A" GOSUB 560
490 ON KEY# 2,"PT B" GOSUB 570
500 ON KEY# 3,"PT C" GOSUB 580
510 ON KEY# 4,"PT D" GOSUB 590
520 RETURN
530 GOSUB 550
540 GOTO 640
550 K1=1 @ K2=2 @ K3=3 @ K4=4 @
K5=5 @ K6=6 @ K7=7 @ K8=8 @
RETURN
560 K1=0 @ RETURN
570 K2=0 @ RETURN
580 K3=0 @ RETURN
590 K4=0 @ RETURN
600 K5=0 @ RETURN
610 K6=0 @ RETURN
620 K7=0 @ RETURN
630 K8=0 @ RETURN
640 CLEAR @ DISP @ DISP "COMMAND
SUMMARY:"
650 DISP
660 DISP "DP100 TAPE MOVEMENT:"
670 DISP " STOP"
680 DISP " REWIND"
690 DISP " FASTFORWARD"
700 DISP
710 DISP "DATA HANDLING:"
720 DISP " TRANSFER"
730 DISP " DIRECTPLOT"
740 DISP " RECORDPLOT"
750 DISP " PICKDUMP"
760 DISP
770 DISP "COMMAND";
780 INPUT C$
790 IF C$="STOP" THEN GOSUB 980
800 IF C$="REWIND" THEN GOSUB 99
0
810 IF C$="FASTFORWARD" THEN GOS
UB 1040
820 IF C$="TRANSFER" THEN GOSUB
1050
830 IF C$="RECORDPLOT" THEN GOSU
B 1730
840 IF C$="DIRECTPLOT" THEN GOSU
B 1690
850 IF C$="PICKDUMP" THEN GOSUB
4370
860 GOTO 640
870 A$(D8,D8)=" "
880 GOSUB 980
890 GOSUB 970
900 CONTROL 4,4 ; 64+4,10
910 L=DP100(A$(1,D8))
920 RESET 4
930 CONTROL 4,4 ; 192+4,10
940 GOSUB 1390
950 Z9=-1
960 RETURN
970 A=1 @ GOTO 1000
980 A=16 @ GOTO 1000
990 A=2
1000 OUTPUT 4 USING "#,K" ; CHR$(
A)&CHR$(A)
1010 WAIT 20
1020 OUTPUT 4 USING "#,K" ; CHR$(
0)&CHR$(0)
1030 RETURN
1040 A=32 @ GOTO 1000
1050 GOSUB 870
1060 GOSUB 1130

```

```

1070 CLEAR @ DISP "PLOT(Y/N)";@
      INPUT Y$
1080 Y$=Y$[1,1]
1090 IF Y$="Y" THEN 1740
1100 ASSIGN# 1 TO *
1110 RETURN
1120 CAT
1130 DISP "FILENAME" @ INPUT F$
1140 L1=CEIL(L/250)+1
1150 ON ERROR GOTO 1120
1160 CREATE F$,L1
1170 OFF ERROR
1180 ASSIGN# 1 TO F$
1190 PRINT# 1,1 ; L,L1,Z9
1200 FOR L2=2 TO L1-1
1210 PRINT# 1,L2 ; A$[L2*250-499
      ,L2*250-250]
1220 NEXT L2
1230 PRINT# 1,L1 ; A$[L1*250-499
      ,L1]
1240 RETURN
1250 CAT
1260 DISP "FILENAME" @ INPUT F$
1270 ON ERROR GOTO 1250
1280 ASSIGN# 1 TO F$
1290 OFF ERROR
1300 READ# 1,1 ; L,L1,Z9
1310 IF L<=D8 THEN 1330
1320 L=D8 @ L1=CEIL(L/250)
1330 FOR L2=2 TO L1-1
1340 READ# 1,L2 ; A$[L2*250-499,
      L2*250-250]
1350 NEXT L2
1360 READ# 1,L1 ; A$[L1*250-499,
      L1]
1370 GOSUB 1390
1380 RETURN
1390 D=FNZ1(1)*100+FNZ1(2)*10+FN
      Z1(3)
1400 H=FNZ1(4)*10+FNZ1(5)
1410 M=FNZ1(6)*10+FNZ1(7)
1420 S=FNZ1(8)*10+FNZ1(9)
1430 N=FNZ1(10)*100+FNZ1(11)*10+
      FNZ1(12)
1440 E=FNZ1(13)*10+FNZ1(14)
1450 C1=FNZ1(15)
1460 P=FNZ1(16)
1470 RESTORE 120
1480 P1=C1*4+P-3
1490 FOR P2=1 TO P1
1500 READ P3
1510 NEXT P2
1520 P=13*P3/1000000
1530 C2=-20/C1
1540 C3=10-C2/2
1550 C4=C1*2
1560 W1=(L-32) DIV C4
1570 W2=W1*C4
1580 RETURN
1590 DEF FNZ(Z1)
1600 FNZ=NUM(A$[Z1,Z1])*256+NUM(
      A$[Z1+1,Z1+1])
1610 FN END
1620 DEF FNZ1(Z2)
1630 Z2=(Z2-1) MOD 16
1640 Z3=Z2 DIV 2*2
1650 Z4=FNZ(Z3+1) DIV 8
1660 IF Z3=Z2 THEN Z4=Z4 DIV 16
1670 FNZ1=Z4 MOD 16
1680 FN END
1690 GOSUB 870
1700 F$=""
1710 GOSUB 1870
1720 RETURN
1730 GOSUB 1260
1740 GOSUB 1870
1750 ASSIGN# 1 TO *
1760 RETURN
1770 O=(C*C2+C3)*C5
1780 J=C*2+J1
1790 J8=C*2-2
1800 PENUP
1810 GRAPH
1820 A=DPC(A$[J,J])
1830 PLOT J-J8,A*F+O
1840 J=J+C4
1850 IF J<J2 THEN 1820
1860 RETURN
1870 CLEAR
1880 DISP "PLOT,PICK OR EXIT";@
      INPUT Y$
1890 IF Y$="PLOT" THEN 1930
1900 IF Y$="PICK" THEN 2410
1910 IF Y$="EXIT" THEN RETURN
1920 GOTO 1870
1930 C5=1
1940 C6=C1
1950 C7=1
1960 DISP "MAGNIFICATION";@ INPU
      T F
1970 DISP "SECONDS PER CM";@ INP
      UT I1
1980 GOSUB 2000
1990 GOTO 2200
2000 PRINT @ PRINT
2010 PRINT "DAY - ";D
2020 T1$=VAL$(H+100)
2030 T$=T1$[2,3]&":"
2040 T1$=VAL$(M+100)
2050 T$=T$&T1$[2,3]&":"
2060 T1$=VAL$(S+100)
2070 T$=T$&T1$[2,3]
2080 PRINT "TIME OF FIRST MARK -
      ";T$
2090 PRINT "EVENT #";E
2100 PRINT USING 2110 ; "PERIOD
      (MS)";P*1000
2110 IMAGE K,X,DDD.DD
2120 PRINT "ID #";N
2130 PRINT USING 2140 ; I1;" SEC
      OND(S) PER CM"

```

```

2140 IMAGE DD,DD,X,K
2150 PRINT "# OF CHANNELS";C1
2160 PRINT USING 2110 ; "VOLTS P
ER DIVISION";2/F
2170 PRINT USING 2110 ; "EVENT L
ENGTH (SECS)";W1*P
2180 PRINT
2190 RETURN
2200 GCLEAR @ GRAPH
2210 I2=INT(9.61*I1/P+.5)
2220 FOR I=1 TO W1 STEP I2
2230 J3=I2
2240 J4=I
2250 GOSUB 2310
2260 COPY
2270 NEXT I
2280 ALPHA
2290 GOTO 1870
2300 REM CONSTANT IN FOLLOWING L
INE SHOULD BE 9, BUT MUST B
E 11 FOR EARLY DP100'S
2310 J1=11+C4*J4
2320 GCLEAR
2330 J2=J1+2+C4*MIN(J3,W1-J4)
2340 J9=J1+2+C4*J3
2350 SCALE J1+2,J9,-10,10
2360 YAXIS J1+2,2
2370 FOR C=C7 TO C6
2380 GOSUB 1770
2390 NEXT C
2400 RETURN
2410 T0=Z9
2420 DISP "SPECIFY FILE FOR SAVI
NG POINTS"
2430 DISP "FILE";@ INPUT F2$
2440 ON ERROR GOTO 2480
2450 ASSIGN# 2 TO F2$
2460 OFF ERROR @ GOTO 2550
2470 CAT
2480 DISP "NEW FILE" @ DISP
2490 OFF ERROR @ DISP "NUMBER OF
SETS OF PICKED POINTS IN
NEW FILE";@ INPUT N9
2500 ON ERROR GOTO 2470
2510 CREATE F2$,CEIL(N9)+1
2520 OFF ERROR
2530 ASSIGN# 2 TO F2$
2540 PRINT# 2,1 ; 0,N9
2550 P1=1 @ P9=W1
2560 B1=-1 @ B2=-1 @ B3=-1 @ B4=
-1
2570 L1$="" @ L2$="" @ L3$="" @
L4$=""
2580 X=128 @ Y=128
2590 GCLEAR @ ALPHA @ CLEAR
2600 DISP "CHANNEL (1,2,3,ALL)";
@ INPUT Y$
2610 IF Y$="1" OR Y$="2" OR Y$="
3" THEN 2650
2620 IF Y$<>"ALL" THEN 2600
2630 C7=1 @ C6=C1 @ C5=1
2640 GOTO 2680
2650 C7=VAL(Y$)
2660 C6=C7
2670 C5=0
2680 J3=P9-P1+1
2690 J4=P1
2700 DISP "MAGNIFICATION";@ INPU
T F
2710 GOSUB 2310
2720 SCALE 0,255,0,191
2730 GCURSOR X,Y,1,10
2740 IF Z9<>-1 THEN 3220
2750 CLEAR
2760 DISP "YOU MUST SPECIFY A TI
ME MARK"
2770 DISP " WINDOW"
2780 DISP "YOU MAY SPECIFY AN IN
TEREST "
2790 DISP " WINDOW"
2800 R1=-1 @ R2=-1
2810 GOSUB 350
2820 KEY LABEL
2830 GOSUB 550
2840 GOSUB 4630
2850 IF NOT K1 THEN 2950
2860 IF NOT K2 THEN 2960
2870 IF NOT K3 THEN 2970
2880 IF NOT K4 THEN 2980
2890 IF NOT K5 THEN 3030
2900 IF NOT K6 THEN GOSUB 2920
2910 GOTO 2840
2920 GRAPH
2930 GOSUB 550
2940 RETURN
2950 L$="(" @ R1=B @ GOTO 2990
2960 L$=")" @ R2=B @ GOTO 2990
2970 L$="[" @ P1=B @ GOTO 2990
2980 L$="]" @ P9=B
2990 MOVE X-3,0 @ LABEL " "
3000 MOVE X-3,0 @ LABEL L$
3010 GCURSOR OFF @ GCURSOR X,Y,1
,10
3020 GOTO 2830
3030 IF R1<>-1 AND R2<>-1 THEN 3
070
3040 CLEAR
3050 DISP @ DISP "YOU MUST SPECI
FY BOTH TIME BRACKET
S"
3060 GOTO 2820
3070 J3=R2-R1+1
3080 J4=R1
3090 GCURSOR OFF
3100 GOSUB 2310
3110 SCALE 0,255,0,191
3120 X=128 @ Y=128
3130 GCURSOR X,Y,1,10
3140 CLEAR
3150 DISP "POSITION CURSOR FOR T
IME MARK AND HIT K1 WHE
N READY"

```

```

3160 GOSUB 550
3170 IF K1 THEN 3170
3180 GOSUB 4630
3190 GOSUB 4160
3200 K2=0
3210 GOTO 3230
3220 GOSUB 550
3230 GOSUB 260
3240 IF NOT K5 THEN 3490
3250 IF NOT K2 THEN 2560
3260 IF NOT K3 THEN 3340
3270 IF NOT K8 THEN 3540
3280 IF NOT K6 THEN 2550
3290 IF NOT K1 THEN GRAPH @ GOTO
      3220
3300 GOTO 3240
3310 GOTO 3240
3320 GOTO 3240
3330 !
3340 I1=J3*P/9.55
3350 GOSUB 2000
3360 PRINT USING 2110 ; "PLOT WI
      NDOW LENGTH : ",J3*P
3370 PRINT "ON FILE ";F$
3380 PRINT "PICKED POINTS ON FIL
      E ";F2$
3390 IF C6<>C7 THEN L9$="ALL" EL
      SE L9$=VAL$(C7)
3400 PRINT "CHANNEL(S) ";L9$
3410 PRINT
3420 GRAPH
3430 COPY
3440 GOTO 3220
3450 !
3460 !
3470 !
3480 !
3490 GCURSOR OFF
3500 ALPHA
3510 ASSIGN# 2 TO *
3520 CLEAR
3530 GOTO 1870
3540 ALPHA
3550 CLEAR
3560 GOSUB 4630
3570 GOSUB 440
3580 KEY LABEL
3590 DISP "TYPE?"
3600 GOSUB 550
3610 IF NOT K5 THEN 3700
3620 IF NOT K6 THEN 4260
3630 IF NOT K7 THEN 4130
3640 IF NOT K8 THEN 3720
3650 IF NOT K1 THEN 4050
3660 IF NOT K2 THEN 4070
3670 IF NOT K3 THEN 4090
3680 IF NOT K4 THEN 4110
3690 GOTO 3610
3700 P1=B @ L$="C"
3710 GOTO 3740

3720 P9=B @ L$="J"
3730 GOTO 3740
3740 CLEAR @ DISP @ DISP
3750 IF B1<>-1 THEN DISP "A";T0+
      B1*P
3760 IF B2<>-1 THEN DISP "B";T0+
      B2*P
3770 IF B3<>-1 THEN DISP "C";T0+
      B3*P
3780 IF B4<>-1 THEN DISP "D";T0+
      B4*P
3790 MOVE GCURSOR X-3,0
3800 LABEL " "
3810 MOVE GCURSOR X-3,0
3820 LABEL L$
3830 GCURSOR OFF @ GCURSOR X,Y,1
      ,10
3840 GOTO 3220
3850 DEF FNP1(Z)
3860 IF Z>256 THEN Z=256
3870 IF Z<1 THEN Z=1
3880 P2=J4-1
3890 P3=J3/256
3900 P4=P3/-2
3910 P4=CEIL(P3*Z+P4+P2)
3920 FNP1=P4
3930 FN END
3940 DEF FNP2(Z)
3950 IF Z>J3+J4-1 THEN Z=J3+J4-1
3960 IF Z<J4 THEN Z=J4
3970 P3=256/J3
3980 P4=P3/2
3990 FNP2=CEIL(P3*(Z-J4)+P4)
4000 FN END
4010 DEF FNP3(P3)
4020 P4=(P3*C1+C7)*2+11
4030 FNP3=DPC(A$[P4,P4])
4040 FN END
4050 B1=B @ L$="A"
4060 DISP "POINT NAME " @ INPUT
      L1$@ GOTO 3740
4070 B2=B @ L$="B"
4080 DISP "POINT NAME " @ INPUT
      L2$@ GOTO 3740
4090 B3=B @ L$="C"
4100 DISP "POINT NAME " @ INPUT
      L3$@ GOTO 3740
4110 B4=B @ L$="D"
4120 DISP "POINT NAME " @ INPUT
      L4$@ GOTO 3740
4130 GOSUB 4160
4140 L$="T"
4150 GOTO 3740
4160 T0=((((D-1)*24+H)*60+M)*60+S
4170 T0=T0-P*B
4180 DISP "TYPE THE INTEGER FOR
      THE TIME          MARK YOU HAVE
      SELECTED."
4190 DISP "  1 - FIRST TIME MARK
           2 - SECOND TIM
           X - XTH T
           E MARK
           IME MARK"

```

```

4200 INPUT Z9
4210 IF Z9<>Z9\1 THEN 4180
4220 T0=T0+2*Z9
4230 Z9=T0
4240 IF LEN(F$)<>0 THEN PRINT# 1
      ,1 ; L,L1,Z9
4250 RETURN
4260 READ# 2,1 ; N8,N9
4270 IF N8<N9 THEN 4330
4280 DISP "FILEFULL"
4290 DISP "HIT CONT TO PROCEED"
4300 PAUSE
4310 L$=""
4320 GOTO 3740
4330 DISP "RECORD DESCRIPTOR ";@
      INPUT B5$
4340 GOSUB 4670
4350 L$=""
4360 GOTO 3740
4370 DISP "SPECIFY FILE WITH STO
      RED POINTS"
4380 GOTO 4400
4390 CAT
4400 DISP "FILE";@ INPUT F2$
4410 ON ERROR GOTO 4390
4420 ASSIGN# 2 TO F2$
4430 OFF ERROR
4440 READ# 2,1 ; N8,N9
4450 PRINT N8;"SETS OF PICKED PT
      S"
4460 PRINT N9;" POSSIBLE"
4470 PRINT "ON FILE - ";F2$
4480 PRINT
4490 FOR N7=2 TO N8+1
4500 GOSUB 4770
4510 PRINT B5$
4520 PRINT USING 4600 ; "A";B1,B
      5,L1$
4530 PRINT USING 4600 ; "B";B2,B
      6,L2$
4540 PRINT USING 4600 ; "C";B3,B
      7,L3$
4550 PRINT USING 4600 ; "D";B4,B
      8,L4$
4560 PRINT "DR100 #";N
4570 PRINT "CHANNEL";C
4580 PRINT
4590 NEXT N7
4600 IMAGE A,X,M8D.4D,X,MOD.DDD,
      X,K
4610 ASSIGN# 2 TO *
4620 RETURN
4630 X=GCURSOR X
4640 Y=GCURSOR Y
4650 B=FNP1(X)
4660 RETURN
4670 PRINT# 2,N8+2 ; 0
4680 IF B1<>-1 THEN PRINT# 2 ; B
      1*P+T0,FNP3(B1),L1$ ELSE PR
      INT# 2 ; -1,-1,"NONE"
4690 IF B2<>-1 THEN PRINT# 2 ; B
      2*P+T0,FNP3(B2),L2$ ELSE PR
      INT# 2 ; -1,-1,"NONE"
4700 IF B3<>-1 THEN PRINT# 2 ; B
      3*P+T0,FNP3(B3),L3$ ELSE PR
      INT# 2 ; -1,-1,"NONE"
4710 IF B4<>-1 THEN PRINT# 2 ; B
      4*P+T0,FNP3(B4),L4$ ELSE PR
      INT# 2 ; -1,-1,"NONE"
4720 PRINT# 2 ; B5$,N
4730 IF C6=C7 THEN PRINT# 2 ; C7
      ELSE PRINT# 2 ; 0
4740 PRINT# 2,1 ; N8+1,N9
4750 N8=N8+1
4760 RETURN
4770 READ# 2,N7 ; L
4780 READ# 2 ; B1,B5,L1$
4790 READ# 2 ; B2,B6,L2$
4800 READ# 2 ; B3,B7,L3$
4810 READ# 2 ; B4,B8,L4$
4820 READ# 2 ; B5$,N,C
4830 RETURN

```

ASSEMBLY LANGUAGE LISTING OF "DPB"

```

1  !*****
2  !*      GCURS BINARY
3  !*      WITH DP100 &DPC
4  !*      (DP100 SERVICE ROUT.)
5  !*****
7      GLO GLOBAL
10     NAM GCURS
20     DEF RUNTIM
30     DEF TOKS
40     DEF PARSES
50     DEF ERMSG
60     DEF INIT
70 RUNTIM BSZ 2
80     DEF GCOFF.
90     DEF GCURX.
100    DEF GCURY.
110    DEF GCURS.
120    DEF REV.
125    DEF DP.
127    DEF DP100.
130 PARSES BSZ 2
140     DEF GCOFFP
150     BSZ 2
160     BSZ 2
170     DEF GCPAR
175     BSZ 2
176     BSZ 2
177     BSZ 2
180 ERMSG BYT 377,377
190 TOKS  ASP "GCURSOR OFF"
200        ASP "GCURSOR X"
210        ASP "GCURSOR Y"
220        ASP "GCURSOR"
230        ASP "REV DATE"
235        ASP "DPC"
236        ASP "DP100"
240        BYT 377
250 !*****
260 INIT  BIN
280        LDBD R34,=ROMFL
290        CMB R34,=2
300        JNZ LOAD?
310 SCRAT! LDM R44,=236,236,236,
236
320        STMD R44,=CHIDLE
330        RTN
340 LOAD?  CMB R#,=5
350        JZR SCRAT!
360 RTN    RTN
370 !*****
380 LEFT  LDM R40,X14,STEP
390
400 RIGHT LDM R40,X14,STEP
410
420 KEY   LDM R14,=BINTAB
430 BIN
440        CMB R16,=2
450        JNZ RTN
460        LDM R22,=KEYHIT
470        CMB R22,=211
480        JZR FRIGHT
490        CMB R22,=223
500        JZR FLEFT
510        CMB R22,=245
520        JZR FUP
530        CMB R22,=242
540        JZR DOWN
550        CMB R22,=234
560        JZR LEFT
570        CMB R22,=235
580        JZR RIGHT
590        CMB R22,=241
600        JZR UP
610        CMB R22,=254
620        JZR FDOWN
630        RTN
640 DOWN  LDM R40,X14,STEP
650        JMP COMDOW
660 UP    LDM R40,X14,STEP
670        JMP COMUP
680 FRIGHT LDM R40,X14,FSTEP
690 COMRIT PUMD R#,+R12
700        LDM R50,X14,CURS-X
710        PUMD R50,+R12
720        JSB =ADDR01
730 COM-X  LDM R40,X14,CURS-Y
740        PUMD R40,+R12
750        JMP COMKEY
760 FLEFT  LDM R40,X14,FSTEP
770 COMLEF LDM R50,X14,CURS-X
780        PUMD R50,+R12
790        PUMD R40,+R12
800        JSB =SUBROI
810        JMP COM-X
820 FUP    LDM R40,X14,FSTEP
830 COMUP  LDM R50,X14,CURS-X
840        PUMD R50,+R12
850        PUMD R40,+R12
860        LDM R40,X14,CURS-Y
870        PUMD R40,+R12
880        JSB =ADDR01
890        JMP COMKEY
900 FDOWN  LDM R40,X14,FSTEP
910 COMDOW LDM R50,X14,CURS-X
920        PUMD R50,+R12
930        LDM R50,X14,CURS-Y
940        PUMD R50,+R12
950        PUMD R40,+R12
960        JSB =SUBROI
970 COMKEY JSB X14,PLOT
980        CLM R50

```

```

990      POMD R40,-R12
1000     PUMD R40,+R12
1010     JSB =COMFLT
1020     POMD R40,-R12
1030     JEN TEST-X
1040     PUMD R40,+R12
1050     LDM R50,=2,0,0,0,0,0
      ,20C,19C
1060     JSB =COMFLT
1070     POMD R40,-R12
1080     JEZ TEST-X
1090     STMD R40,X14,CURS-Y
1100 TEST-X CLM R50
1110     POMD R40,-R12
1120     PUMD R40,+R12
1130     JSB =COMFLT
1140     POMD R40,-R12
1150     JEN MOVCUR
1160     PUMD R40,+R12
1170     LDM R50,=2,0,0,0,0,0
      ,60C,25C
1180     JSB =COMFLT
1190     BIN
1200     POMD R40,-R12
1210     JEZ MOVCUR
1220     STMD R40,X14,CURS-X
1230 MOVCUR JSB X14,PLOT
1240     CLE
1250     JSB =EOJ2
1260     LDBD R31,X14,KEYCON
1270 LOOPKE LDBD R30,=KEYSTS
1280     LRB R30
1290     JEV EOJ
1300     LDBD R30,=CRTSTS
1310     LRB R30
1320     JEV LOOPKE
1330 LOOPK2 LDBD R30,=KEYSTS
1340     LRB R30
1350     JEV EOJ
1360     LDBD R30,=SVCWRD
1370     JOD EOJ
1380     LDBD R30,=CRTSTS
1390     LRB R30
1400     JOD LOOPK2
1410     DCB R31
1420     JNZ LOOPKE
1430     LDB R31,=KYRPT2
1440     STBD R31,X14,KEYCON
1450     LDM R20,=KEY
1460     ADM R20,R14
1470     DCM R20
1480     LDM R4,R20
1490 EOJ    LDB R31,=KYRPT1
1500     STBD R31,X14,KEYCON
1510     POMD R44,-R6
1520     CLE
1530     RTN
1540 *****
*****
*****

1550 GCPAR PUBD R43,+R6
1560     JSB =NUMVA+
1570     JEN OK
1580 ERR    JSB =ERROR+
1590     BYT 81D
1600 OK     JSB =GETCMA
1610     JSB =NUMVAL
1620     JEZ ERR
1630     CMB R14,=54
1640     JNZ DONE
1650     JSB =NUMVA+
1660     JEZ ERR
1670     JSB =GETCMA
1680     JSB =NUMVAL
1690     JEZ ERR
1700 DONE   POBD R47,-R6
1710     LDB R45,=371
1720     PUMD R45,+R12
1730     RTN
1740 GCOFFP PUBD R43,+R6
1750     JSB =SCAN
1760     JMP DONE
1770 *****
*****
*****
1780     BYT 241
1790 GCOFF.  LDMD R14,=BINTAB
1800     JSB X14,SCRAT!
1810     JSB X14,PLOT
1820     RTN
1830 *****
*****
*****
1840     BYT 0,55
1850 GCURX.  LDMD R14,=BINTAB
1860     LDMD R50,X14,CURS-X
1870 GPUSH   PUMD R50,+R12
1880     RTN
1890 *****
*****
*****
1900     BYT 0,55
1910 GCURY.  LDMD R14,=BINTAB
1920     LDMD R50,X14,CURS-Y
1930     JMP GPUSH
1940 *****
*****
*****
1950     BYT 241
1960 GCURS.  BIN
1970     LDMD R14,=BINTAB
1980     LDM R40,=0,0,0,0,0,0
      ,0,10C
1990     STMD R40,X14,STEP
2000     LDB R47,=40C
2010     STMD R40,X14,FSTEP
2020     LDM R20,R12
2030     SBM R20,=40,0
2040     CMMD R20,=TOS
2050     JNZ NOSTEP

```

```

2060      JSB =ONER
2070      BIN
2080      STMD R#,X14,FSTEP
2090      JSB =ONER
2100      BIN
2110      STMD R#,X14,STEP
2120 NOSTEP JSB =ONER
2130      BIN
2140      STMD R#,X14,CURS-Y
2150      JSB =ONER
2160      BIN
2170      STMD R#,X14,CURS-X
2180      JSB X14,PLOT
2190      LDM R46,=KEY
2200      ADM R46,R14
2210      STM R46,R45
2220      LDB R47,=236
2230      LDB R44,=316
2240      STMD R44,=CHIDLE
2250      RTN
2260 *****
*****
*****
2270 PLOT      JSB X14,GCURX.
2280          JSB X14,GCURY.
2290          LDM R20,=ROMTAB
2300 NXTROM   POMD R24,+R20
2310          CMB R24,=377
2320          JZR SYSTEM
2330          CMB R24,=PPROM#
2340          JNZ NXTROM
2350          JSB =ROMJSB
2360          DEF PMOVE.
2370          VAL PPROM#
2380          LDMD R14,=BINTAB
2390          JMP PLOT++
2400 SYSTEM   JSB =MOVE.
2410 PLOT++   LDM R20,=CURSES
2420          ADM R20,R14
2430          LDBD R22,=XMAP
2440          ANM R22,=3,0
2450          LDM R34,R22
2460          LLM R34
2470          LLM R34
2480          ADM R34,R22
2490          ADM R34,R20
2500          LDM R22,=5,0
2510          LDM R44,=1,0,1,0
2520          JSB =BPLOT+
2530          RTN
2540 *****
*****
*****
2550 CURSES   BYT 360,300,240,220,
10
2560          BYT 170,140,120,110,
4
2570          BYT 74,60,50,44,2
2580          BYT 36,30,24,22,1
2590 KEYCON   BSZ 1

2600 CURS-X   BSZ 10
2610 CURS-Y   BSZ 10
2620 FSTEP    BSZ 10
2630 STEP     BSZ 10
2640 *****
*****
*****
2650          BYT 0,56
2660 REV.      BIN
2670          LDM R44,=11D,0
2680          DEF DATE
2690          ADMD R46,=BINTAB
2700          PUMD R44,+R12
2710          RTN
2720 DATE      ASC "SEP 29,1982"
2730 BPLOT+    DAD 34405
2740 MOVE.     DAD 31703
2750 PMOVE.    DAD 64400
2760 ROMJSB    DAD 4776
2770 PPROM#    EQU 360
2780 ROMTAB    DAD 101235
2790 KYRPT2    EQU 1
2800 KYRPT1    EQU 30
2810 CRTSTS    DAD 177406
2820 KEYSTS    DAD 177402
2830 CHIDLE    DAD 102416
2840 ROMFL     DAD 101231
2850 KEYHIT    DAD 100671
2860 EOJ2      DAD 34772
2870 ADDR01    DAD 52150
2880 SUBR01    DAD 52127
2890 BINTAB    DAD 101233
2900 NUMVAL    DAD 12412
2910 NUMVA+    DAD 12407
2920 SCAN      DAD 11262
2930 GETCMA    DAD 13414
2940 SVCWRD    DAD 100151
2950 TOS       DAD 101132
2960 ERROR+    DAD 6611
2970 ONER      DAD 56215
2980 XMAP      DAD 100262
2990 COMFLT    DAD 32621
3000          LNK DP2S

```

```

1000 !*****
1010 !  BINARY PROGS
1020 !    DP100DECODE
1030 !
1040 !
1050 !
1060 !*****
1070 !*****
1080 ! Runtime Routines
1090 !*****
1100         BYT 30,55
1110 DP.     BIN
1120         POMD R24,-R12
1130         POMD R20,-R12
1140         LDM R36,=0,0
1150         POBD R23,+R24
1160         POBD R22,+R24
1170         LDM R36,R22
1180         NCM R36
1190         ANM R36,=377,17
1200         SBM R36,=0,10
1210         JSB =CONBIN
1220         BIN
1230         LDM R50,=97C,99C,00C
           ,00C,25C,81C,82C,48C
1240         JSB =MPY10
1250         RTN
1260 MPY10   DAD 52562
1270 CONBIN  DAD 3572
1280         LNK DP35

```

```

10      BYT 30,55
20  DP100.  BIN
30      LDMD R22,=BINTAB
40      POMD R26,-R12
50      POMD R30,-R12
60      LDM R36,R30
70      JSB =CONBIN
80      LDM R50,R40
90      LDM R32,R26
100     LDM R24,=122,377
110     LDM R26,=123,377
120     JSB X22,INPUT
130  DPDONE  LDM R36,R30
140     BIN
150     JSB =CONBIN
160     JSB =SUB10
170  !      PUMD R40,+R12
180     RTN
190  IBF=1   LDBD R37,R24
200     JEV IBF=1
210     RTN
220  OBF=0   LDBD R37,R24
230     JNG OBF=0
240     RTN
250  DIRCMD  JSB X22,0=B=0
260     LDB R37,=2
270     STBD R37,R24
280     STBD R36,R26
290     JSB X22,OBF=0
300     CLB R37
310     STBD R37,R24
320     RTN
330  0=B=0   LDBD R37,R24
340     ANM R37,=202
350     JNZ 0=B=0
360     RTN
370  INPUT   BIN
380     LDB R36,=20
390     JSB X22,DIRCMD
400  IN4      JSB X22,WORDIN
410     JNZ IN2
420     CMB R76,=60
430     JZR IN4
435     JMP IN9
440  IN7      JSB X22,WORDIN
450     JNZ IN2
460  IN9      TSB R76
470     JZR IN5
480     ADM R30,=2,0
490     SBM R32,=2,0
500     JMP IN7
510  ! IN8    LDBD R77,X32,MIN2
520  !      ANM R77,=17
530  !      STBD R77,X32,MIN2
540  IN5      JSB X22,WORDIN
550     JNZ IN2
560     CMB R76,=60
570     JNZ IN5
580     JSB X22,WORDIN
590     JNZ IN2

600     JSB X22,WORDIN
610     JNZ IN2
620     JSB X22,WORDIN
630     JNZ IN2
640     JSB X22,WORDIN
650     JNZ IN2
660     JSB X22,WORDIN
670     JNZ IN2
680     JSB X22,WORDIN
690     JNZ IN2
700     JSB X22,WORDIN
710     JNZ IN2
720  IN2      LDB R37,=4
730     STBD R37,R24
740     RTN
750  SUB10    DAD 52137
755  MIN2     EQU 177776
760  WORDIN   JSB X22,IBF=1
770     LDBD R76,R26
780     NCB R76
790     ANM R76,=77,0
800     PUBD R76,+R32
810     ANM R37,=4
820     JNZ WOREN1
830     DCM R30
840     JZR WOREND
850     STBD R#,R26
860     JSB X22,IBF=1
870     LDBD R77,R26
880     NCB R77
890     PUBD R77,+R32
900     ANM R37,=4
910     JNZ WOREN1
920     DCM R30
930     JZR WOREND
940     STBD R#,R26
950     ANM R76,=60,0
960     CLB R36
970     RTN
980  WOREN1   DCM R30
990  WOREND   LDB R36,=377
1000     RTN
1010     FIN

```

GPIO CABLE WIRE ASSIGNMENTS

<u>DB25S PIN</u>	<u>NAME</u>	<u>WIRE COLOR</u>	<u>HP-85 PIN</u>	<u>HP-85 NAME</u>
<u>Cable A:</u>				
18	GROUND	BLACK	1	GND
5	FAST	WHITE/BLUE	2	DC5
		WHITE/GREY	3	DC7
		WHITE/ORANGE	4	DC2
16	PLAY	WHITE/BROWN	5	DC0
		WHITE/BLK/GRN	6	ST0
		WHITE/BLK/RED	7	OUTAL
		WHITE/BLACK	8	CTLA
22	B7	GREY	9	DA7
23	B5	BLUE	10	DA5
13	B0	BROWN	11	DA0
12	B2	ORANGE	12	DA2
		WHITE	13	GND
4	STOP	WHITE/GREEN	14	DC4
		WHITE/VIOLET	15	DC6
		WHITE/YELLOW	16	DC3
17	REWIND	WHITE/RED	17	DC1
6	STB	WHITE/BLK/YEL	18	FLGA
		WHITE/BLK/ORNG	19	RESAL
		WHITE/BLK/BRN	20	CTL0
10	B6	VIOLET	21	DA6
11	B4	GREEN	22	DA4
25	B1	RED	23	DA1
24	B3	YELLOW	24	DA3
<u>Cable B:</u>				
18	GROUND	BLACK	1	GND
		WHITE/BLUE	2	DD5
		WHITE/GREY	3	DD7
		WHITE/ORANGE	4	DD2
		WHITE/BROWN	5	DD0
		WHITE/BLK/GRN	6	ST1
		WHITE/BLK/RED	7	OUTBL
		WHITE/BLACK	8	CTLB
18	GROUND	GREY	9	DB7
19	B13	BLUE	10	DB5
9	B8	BROWN	11	DB0
8	B10	ORANGE	12	DB2
		WHITE	13	GND
		WHITE/GREEN	14	DD4
		WHITE/VIOLET	15	DD6
		WHITE/YELLOW	16	DD3
		WHITE/RED	17	DD1
		WHITE/BLK/YEL	18	FLGA
		WHITE/BLK/ORNG	19	RESBL
		WHITE/BLK/BRN	20	CTL1
18	GROUND	VIOLET	21	DB6
7	B12	GREEN	22	DB4
21	B9	RED	23	DB1
20	B11	YELLOW	24	DB3

INDEX

CURSOR CONTROL KEYS	12
DATA HANDLING COMMANDS	6
DIRECTPLOT	8
EXAMPLE PLOT	10
GPIO WIRE ASSIGNMENTS	33
HARDWARE SETUP	4
NEW DP-100	4
OLD DP-100	4
PICK	9, 11, 13
PICK - USER SEQUENCE	17
PICK NOTES	11
PICKDUMP	8
PICKDUMP OUTPUT	9
PLOT	9, 10
PROGRAM STARTUP	5
PROGRAM LISTINGS	21, 27
RECORDPLOT	8
SPECIAL FUNCTION KEYS	11
TAPE MOVEMENT COMMANDS	6
TRANSFER	7